Some Maple Commands for Doing Division with Remainder

[This goes with the Factorization Theory handout.]

```
> restart;
  with(GaussInt):
```

GIgcdex runs the Extended Euclidean Algorithm on Gaussian integers.
Here's our example from the handout:

```
> GIgcdex(-1+5*I,-3+I,'s','t');

  's' = s;
  't' = t;

  s*(-1+5*I)+t*(-3+I);
```

$$1+I$$
$$s=1$$
$$t=-1+I$$
$$1+I \quad \textbf{(1)}$$

So for this example we get: s(-1+5i) + t(-3+i) = gcd = 1+i
Let's recompute using repeated division with remainder (we'll run the algorithm ourselves)...
GIquo is the Gaussian integer division with remainder command.

```
>  q_1 := GIquo(-1+5*I,-3+I,'r_1');
  'r_1' = r_1;
```

$$q\_1 := 1-I$$
$$r\_1 = 1+I \quad \textbf{(2)}$$

Our first remainder is 1+i (non-zero).

```
>  q_2 := GIquo(-3+I,r_1,'r_2');
  'r_2' = r_2;
```

$$q\_2 := -1+2I$$
$$r\_2 = 0 \quad \textbf{(3)}$$

Our second remainder is 0, so the last non-zero remainder (i.e. 1+i) must be the GCD.
Backtracking gives us the "extended" result...

```
> r_1 = (-1+5*I)+(-q_1)*(-3+I);
```

$$1+I = 1+I \quad \textbf{(4)}$$

Now let's try the algorithm with regular polynomials. The "rem" command divides polynomials storing the quotient in a specified variable and spitting back the remainder.

x^3-2 divided by x^2+3x+2 is...

```
> rem(x^3-2,x^2+3*x+2,x,'q');
```

$$7x+4 \quad \textbf{(5)}$$

```
> q;
```

$$x - 3 \quad (6)$$

We get that the quotient is $x-3$ and the remainder is $7x+4$.
Let's go again...

```
> rem(x^2+3*x+2,7*x+4,x,'q');
```

$$\frac{30}{49} \quad (7)$$

```
> q;
```

$$\frac{1}{7}x + \frac{17}{49} \quad (8)$$

This time the quotient is $\frac{1}{7}x + \frac{17}{49}$ and the remainder is $\frac{30}{49}$.
If we divide again, we'll get a remainder of 0 (since 30/49 is a unit). This means that the GCD of x^3-2 and x^2+3x+2 is 30/49 (or normalizing, the GCD = 1).
Running the algorithm backwards yields...

```
> simplify((-x/7-17/49)*(x^3-2)+(x^2/7-4/49*x-2/49)*(x^2+3*x+2));
```

$$\frac{30}{49} \quad (9)$$

So the desired $s(x)$ and $t(x)$ such that $s(x)\,(x^3-2)+t(x)\,(x^2+3x+2)=1$ are...

```
> s(x) = 49/30*(-x/7-17/49);
```

$$s(x) = -\frac{7}{30}x - \frac{17}{30} \quad (10)$$

```
> t(x) = 49/30*(x^2/7-4/49*x-2/49);
```

$$t(x) = \frac{7}{30}x^2 - \frac{2}{15}x - \frac{1}{15} \quad (11)$$

This means that $\frac{7}{30}x^2 - \frac{2}{15}x - \frac{1}{15}$ represents the inverse of x^2+3x+2 in the quotient ring $\frac{\mathbb{Q}[x]}{(x^3-2)}$.
Let's check...

```
> rem((x^2+3*x+2)*(7/30*x^2-2/15*x-1/15),x^3-2,x,'q');
```

$$1 \quad (12)$$

Notice that Maple can "rationalize" just as well as we can...

```
> 1/(2^(2/3)+3*2^(1/3)+2);
```

$$\frac{1}{2^{2/3}+3\,2^{1/3}+2} \quad (13)$$

```
> rationalize( 1/(2^(2/3)+3*2^(1/3)+2) );
```

$$\frac{7}{30}2^{2/3} - \frac{2}{15}2^{1/3} - \frac{1}{15} \quad (14)$$